
Techniques for Programming the HP 892X Family of Instruments

Product Note



**Key programming
techniques**

Table of Contents

Basic Input and Output	3
Using Status Registers for Cellular Call Processing	5
Triggering	8
Error Handling.....	9
Reducing Test Time	10

Introduction

This product note describes some key programming techniques for the HP 892X family of products. This document can be used as a training tool for users that are new to programming test equipment, or it can be used by experienced programmers that need to know about unique programming required for the HP 892X family of instruments.

The HP 892X family of products refers to HP 8920, 8921, 8922, and 8924. Most topics in this document apply to all of these products, however the examples on cellular call processing are specific to the HP 8920A/B and HP 8924C.

To simplify this document, the discussion and examples are written using HP BASIC on an external computer. For programmers using languages other than HP BASIC, the concepts and HP-IB command syntax will still be important and helpful.

There are many documents from Hewlett-Packard that are useful to learn about programming the HP 892X family of instruments. Some suggested documents are:

- *HP 8920/21 Programmers Guide*, part number: 08920-90204
- *Manual and Automated IS-98 Testing with the HP 8924C* (Also useful for HP 8920A/B cellular programming) part number: 5965-4775E
- *HP-IB Instrument Programming Using HP Series 200/300 BASIC*, part number: 50011-60032

Your comments and feedback on this document are important to Hewlett-Packard. Please e-mail or FAX your suggestions to:

Allen Henley
e-mail: henley@spk.hp.com
FAX 1-509-921-3700

Basic Input and Output from the HP 892X

Sometimes the most difficult part of controlling the test set is just getting started with your first measurement. Before attempting to automate a measurement, it is generally recommended to first perform the set up and make measurements manually, then duplicate the manual procedure with HP-IB command strings and programming algorithms.

There are two fundamental processes for communicating with the HP 892X over the HP-IB bus. The first process is sending data (like commands) to the instrument. In HP BASIC, the command to send data to the instrument is the OUTPUT statement. The second process for communicating with the HP 892X is reading data back from the instrument. In HP BASIC, the command to read data from the instrument is the ENTER statement.

Examples for using the OUTPUT statement:

An example two line program for sending a command to the HP 892X is:

```
10 OUTPUT 714; “*RST”
20 END
```

Explanation:

1. Line 10 sends a command to the instrument at address 714. The command is “*RST” which instructs the instrument to reset.
2. Line 20 is the END statement which is always required in any HP BASIC program. This line has no effect on the instrument.

To perform the initial set up for measurements on the HP 892X, the general process is:

- Use the OUTPUT statement to display the correct screen.
- Use the OUTPUT statement for making the settings on that screen.

An example of this would be:

```
10 OUTPUT 714; “DISP RFG”
20 OUTPUT 714; “RFG:FREQ 825 MHZ”
30 END
```

Explanation:

1. Line 10 displays the RF Generator Screen.
2. Line 20 sets the RF Generator frequency to 825 MHz.

This process is repeated over and over until all the necessary settings are made.

When the instrument settings are made and the radio under test is configured, the next step is to read a measurement from the instrument. In HP BASIC the ENTER statement is used to read data from an instrument. The ENTER statement is always preceded by an OUTPUT statement that tells the instrument which measurement the program wants to read.

Examples for using the ENTER statement:

```
10 OUTPUT 714;“MEAS:RFR:POW?”
20 ENTER 714; Data
30 END
```

Explanation:

1. Line 10, the OUTPUT statement is used to instruct the instrument to measure RF power. This line contains a question mark “?” and is called a query.
2. Line 20 is the ENTER statement that will actually read the result back from the HP 892X. The program will hang on line 20 until the test set responds with data. You must always follow a query with an ENTER statement to read the data back from the instrument. You should include some type of time out statement in your program to handle the situation when the test set cannot make the measurement.

The basic concept of reading measurements from the HP 892X is quite simple, however there are many factors that makes reading data one of the most difficult automated processes. When making measurements you need to consider measurement settling, measurement triggering, and error handling if the measurement doesn't complete. An automated procedure for setting up and making measurements should include the following steps:

1. Configure the HP 892X and radio.
2. Display and Activate the measurement on the HP 892X.
3. Wait until the test set and radio has “settled”.
4. Trigger the measurement.
5. Output a query to the test set to tell it which result you want.
6. Enter the data back from the test set.
7. If the data doesn't come back from the test set, send a device clear to the test set so that the query is not still pending.

If any of these steps are not done properly you will receive “Query Error” messages on the HP 892X and it is possible to lockup the test set if errors are not handled correctly.

A more complete example that includes everything except error handling is the following:

```
10 OUTPUT 714;"TRIG:MODE:RETR SINGLE"  
20 OUTPUT 714;"DISP RFG"  
30 OUTPUT 714;"RFG:FREQ 825 MHZ"  
40 OUTPUT 714;"RFG:AMPL -10 DBM"  
50 OUTPUT 714;"DISP RFAN"  
60 OUTPUT 714;"MEAS:RFR:POW:STATE ON"  
70 WAIT 2  
80 OUTPUT 714;"TRIG"  
90 OUTPUT 714;"MEAS:RFR:POW?"  
100 ENTER 714;Data  
110 PRINT Data  
120 END
```

Explanations:

1. Line 10 sets the instrument for single triggering.
2. Line 20 displays the RF Generator screen to allow settings.
3. Lines 30 and 40 set the RF Generator frequency and amplitude.
4. Line 50 displays the RF Analyzer screen.
5. Line 60 enables the RF power measurement. Once the measurement is enabled, it will remain active when ever a screen is displayed that contains the RF power measurement.
6. Line 70 is a wait statement that allows settling for the HP 892X and the radio under test
7. Line 80 triggers the measurement.
8. Line 90 instructs the instrument to put the RF power result in the output queue to be read back by the controlling program.
9. Line 100 enters the data from the HP 892X output queue.
10. Line 110 prints the result to the computer screen.

More details on triggering and error handling are included later in this document.

Using Status Registers for Cellular Call Processing

Cellular call processing refers to registrations, originations, pages, and handoffs for cellular mobiles. This section applies to the HP 8920A/B and 8924C. These test sets contain built in call processing to perform all these cellular functions. Because executing these functions will take an unknown amount of time, it is necessary for the control program to use special techniques for monitoring the state of the test set and taking special care to keep the control program synchronized with the test set. This is accomplished by monitoring the status registers which indicate the current state of the test set. Each bit of the status registers corresponds to a unique state or process in the instrument. For a very detailed explanation of the capabilities, refer to the HP 8920/21 Programmers Guide, part number: 08920-90204.

The concept of writing programs that use status registers is quite simple, but since many users have never programmed the status registers it may seem a little difficult at first. The general procedure is to execute a call processing function (like maybe a PAGE), and then monitor one of the bits in the call processing status register group until the bit indicates that the function has completed. For instance, when executing a page, your control program can monitor the “connected” bit to indicate that the page was successful and the control program can continue.

In these programming examples the call processing condition register is monitored. This register indicates the current instantaneous state of the call processing system. In some programming situations it desirable to monitor the call processing event register. The event register latches changes in the condition register.

A brief programming example will help:

```
10 OUTPUT 714;“DISP ACNT”
20 OUTPUT 714;“*CLS”
30 OUTPUT 714;“CALLP:PAGE”
40 REPEAT
50   OUTPUT 714;“STAT:CALLP:COND?”
60   ENTER 714;Status_register
70   WAIT .2
80 UNTIL BIT(Status_register,5)
90 PRINT “Page was Successful!”
100 END
```

Explanations:

1. Line 10 displays the Analog Call Control screen. This is the screen where call processing functions take place.
2. Line 20 is a command to “Clear Status Registers”. This is important because the program will monitor these registers to determine the instrument state.
3. Line 30 commands the HP 892X to PAGE the mobile.
4. Lines 40 and 80 create a loop to monitor the call processing status condition register.
5. Line 50 queries the HP 892X for the call processing status condition register.
6. Line 60 reads the value back from the test set and stores the data in a variable called Status_register.
7. Line 70 is a short wait (200 milliseconds) which allows the test set some time to process other tasks. This is important because the HP-IB queries have a high priority and if the loop doesn’t have any delays, the test set is so busy reporting to the external computer that the other processes (like the page) may not complete.
8. Line 80 examines bit 5 of the variable Status_register. When bit 5 is set, it indicates the mobile is connected and only then should the control program continue on and make measurements.

This short routine to page the mobile is complete except for any error trapping. If the mobile doesn’t connect, the program will stay in a continuous loop. There are two techniques that are commonly used for this situation.

The first technique involves monitoring the “Event Status Register”, which is where error conditions are reported. The second technique involves putting a timer into the loop and if the call processing function doesn’t complete in a specified time, the control program can exit the loop and continue with some type of error trapping routine.

The following is an example of the technique that monitors the Event Status Register for errors:

```
10  OUTPUT 714;"DISP ACNT"  
20  OUTPUT 714;"*CLS"  
30  OUTPUT 714;"CALLP:PAGE"  
40  REPEAT  
50    OUTPUT 714;"STAT:CALLP:COND?"  
60    ENTER 714;Status_register  
70    OUTPUT 714;"*ESR?"  
80    ENTER 714;Error_register  
90    IF Error_register<>0 THEN  
100   PRINT "Error during page"  
110   STOP  
120   END IF  
130   WAIT .2  
140  UNTIL BIT(Status_register,5)  
150  PRINT "Page was Successful!"  
160  END
```

Explanations:

1. Lines 70 and 80 are used to read the event status register.
2. Lines 90 checks if the variable Error_register is not equal to zero, which would indicate an error.
3. Lines 100-120 prints an error message and stops the program.

The following is an example of the technique that uses a loop counter as a timer to detect the page failure:

```
10  ! re-save "/nfs/hpspkeb/users_dir4/henley/rmb/  
    892xprog/page3.pgm"  
20  OUTPUT 714;"DISP ACNT"  
30  OUTPUT 714;"*CLS"  
40  OUTPUT 714;"CALLP:PAGE"  
50  Loop_count=0  
60  REPEAT  
70  Loop_count=Loop_count+1  
80  OUTPUT 714;"STAT:CALLP:COND?"  
90  ENTER 714;Status_register  
100 IF Loop_count=50 THEN  
110 PRINT "Error during page"  
120 STOP  
130 END IF  
140 WAIT .2  
150 UNTIL BIT(Status_register,5)  
160 PRINT "Page was Successful!"  
170 END
```

The process to use status registers for performing registrations, pages, originations, and handoffs are all similar. The following example shows how to perform a registration. One small difference you may notice in this example is the monitoring loop is looking for the HP 892X to leave the registration condition (indicating the registration is finished). In the example program for a page, the monitoring loop was waiting for the HP 892X to enter the connected condition (indicating that the page state was finished).

```
10 ! re-save "/nfs/hpspkcb/users_dir4/henley/rmb/
    892xprog/reg.pgm"
20 OUTPUT 714;"DISP ACNT"
30 OUTPUT 714;"*CLS"
40 OUTPUT 714;"CALLP:REGISTER"
50 WAIT 2
60 Loop_count=0
70 REPEAT
80     Loop_count=Loop_count+1
90     OUTPUT 714;"STAT:CALLP:COND?"
100    ENTER 714;Status_register
110    IF Loop_count=50 THEN
120    PRINT "Error during registration"
130    STOP
140    END IF
150    WAIT .2
160 UNTIL NOT BIT(Status_register,1)
170 PRINT "Registration was Successful!"
180 END
```

Explanations:

1. Line 20 displays the Analog Call Control screen. This is the screen where call processing functions take place.
2. Line 30 is a command to "Clear Status Registers". This is important because the program will monitor these registers to determine the instrument state.
3. Line 40 commands the HP 892X to REGISTER the mobile.
4. Line 50 is a two second wait that allows the test set to enter the "registration" state.
5. Lines 60 to 160 create a loop to monitor the call processing status condition register.
6. Line 80 is a loop counter to know when to exit because of a time out.
7. Line 90 and 100 queries the HP 892X for the call processing status condition register.
8. Lines 110-140 checks the loop counter and stops the program if the count = 50.
9. Line 150 is a short wait (200 milliseconds) which allows the test set some time to process other tasks.
10. Line 160 monitors the "registration bit" until the HP 892X leaves the registration state. This is indicated by the bit being set to zero (false).

Triggering Measurements

There are two types of triggering modes for remote (automated) operation. When the HP 892X is in remote mode, the trigger is set to either repetitive (continuous) or single. In repetitive triggering mode, all active measurements are continuously triggered. When the control program queries the HP 892X for a measurement result, the HP 892X triggers a new measurement and places the last previously completed measurement result in the output queue for the control program to read.

When using single trigger, the control program must send a "TRIG" command to the HP 892X. This flushes all the current measurement results and causes the HP 892X to begin a new measurement. The new measurement will update the results for all the active measurements. After the measurement results are updated, the control program can query the HP 892X for results. This causes the HP 892X to place the requested results in the output queue where the control program can then retrieve the measurement results.

Two short examples will demonstrate the two different triggering modes:

Repetitive Triggering example:

```
10 ! re-save "/nfs/hpspkcb/users_dir4/henley/rmb/
    892xprog/trig.pgm"
20 OUTPUT 714;"TRIG:MODE:RETR REPETITIVE"
30 OUTPUT 714;"DISP RFAN"
40 OUTPUT 714;"MEAS:RFR:POW?"
50 ENTER 714;Result
60 PRINT Result
70 END
```

Explanation:

1. Line 20 sets the remote triggering mode to repetitive (continuous).
2. Line 30 displays the RF Analyzer screen. This will automatically start triggering for all the active measurements on that screen.
3. Line 40 queries the test set for the RF Power measurement. This causes the last result to be put into the output queue.
4. Line 50 enters the data from the output queue into the variable called "Result".

Single Triggering example:

```
10 ! re-save "/nfs/hpspkcb/users_dir4/henley/rmb/
    892xprog/trig2.pgm"
20 OUTPUT 714;"TRIG:MODE:RETR SINGLE"
30 OUTPUT 714;"DISP RFAN"
40 OUTPUT 714;"TRIG"
50 OUTPUT 714;"MEAS:RFR:POW?"
60 ENTER 714;Result
70 PRINT Result
80 END
```

Explanation:

1. Line 20 sets the remote triggering mode to single.
1. Line 30 displays the RF Analyzer screen. No measurements will appear.
1. Line 40 triggers all the active measurements on the RF Analyzer screen.
1. Line 50 queries the test set for the RF Power measurement. This causes the result from the previous trigger to be put into the output queue.
1. Line 60 enters the data from the output queue into the variable called "Result".

Handling HP-IB Query Errors

Under certain conditions, it may not be possible for the HP 892X to make a measurement. If this happens, the control program must properly clear the HP-IB buffer before proceeding with any other program statements. A common mistake is when a measurement doesn't complete, the control program is stopped and rerun. Quite often the first command sent to the instrument is "*RST" for an instrument reset. This is not the correct way to handle the query error and it can lockup the test set under certain conditions. The correct technique would be to send a "device clear" to the instrument before any commands are sent. It is a good idea to have the very first statement in any HP-IB control program to be a device clear.

The device clear (or selected device clear) command is defined in IEEE 488 and it instructs the instrument to clear any pending HP-IB operations. The device clear statement is unique for each different interface card and programming language. As an example, in HP BASIC the command "CLEAR 714" sends a selected device clear to the instrument at address 714. In a language like C or Pascal, the command could be "IOCLEAR (714)" or "IOCLEAR (714L)". Refer to the documentation for your programming language and HP-IB or GP-IB interface card.

An example of a good programming technique to start your control program could be:

```
10 CLEAR 714
20 OUTPUT 714;"*RST"
30 END
```

By starting the program with a device clear, any pending HP-IB operations will be cleared and there won't be any chance of locking up the HP 892X. Typically the control program will have some type of error trapping to clean up the HP-IB queue in case of measurement errors or measurements that don't complete. A simple example can illustrate how a programmer could implement this in HP BASIC.

```
10 ! re-save "/nfs/hpspkcb/users_dir4/henley/rmb/
892xprog/err.pgm"
20 Start_meas: !
30 ON TIMEOUT 7,15 GOTO Clean_up_hpib
40 OUTPUT 714;"TRIG:MODE:RETR SINGLE"
50 OUTPUT 714;"DISP RFAN"
60 OUTPUT 714;"TRIG"
70 OUTPUT 714;"MEAS:RFR:POW?"
80 ENTER 714;Result
90 PRINT Result
100 STOP
110 Clean_up_hpib:!
120 CLEAR 714
130 PRINT "Measurement was aborted"
140 GOTO Start_meas
150 END
```

Explanations:

Line 30 is an HP BASIC statement that sets up a timeout. This statement instructs HP BASIC to go to the line labeled "Clean_up_hpib" if any instrument on the HP-IB bus 7 doesn't respond for 15 seconds. (In this example, 7 is the interface select code). If the power measurement doesn't complete, the program will hang up on line 80 for 15 seconds while it is waiting for data back from the HP 892X. After the 15 second "ON TIMEOUT" timer expires, the program jumps to line 110, then executes line 120 to clear the HP-IB operations. Line 130 prints an error message and line 140 causes the program to go to the beginning (Line 20) and starts over.

This is a simple technique that can be implemented in almost any programming language. It is a good programming practice that whenever data is to be sent to the instrument or read from the instrument, some type of error handling routine should be used. One implementation would be to write general purpose subroutines like "To_testset" or "From_testset" that would send data back and forth between the computer and the test set. These general purpose routines would be a good location in the code to include the proper error handling routines.

Reducing Test Time

Writing the fastest possible test software involves efficient programming, understanding the device-under-test, and optimizing set ups and measurements with the test equipment. This section of this document focuses on the test set, and particularly on helping the programmer understand and avoid some of the slower (and often unnecessary) processing in the test set.

The following processes in the HP 892X family are relatively slow compared to automated applications.

- Changing screens when it isn't necessary.
- Auto-ranging and auto-tuning.
- Making unnecessary measurements.

Changing screens on the HP 892X takes approximately one second. For making measurements it is generally necessary to be on the correct screen, however many settings (such as the RF Generator) can be made from almost any screen. For analog measurements, try to use the DUPLEX screen because it contains Generator and Analyzer functions. Avoid the Analog RX and TX screens because they auto-configure many setting in the test set and this is relatively slow.

Auto-ranging and auto-tuning for the analyzer are two processes that should definitely be avoided. Auto-ranging is the process where the test set analyzer automatically set the correct gain and attenuation setting for a measurement. Auto-tuning in the test set is where the RF analyzer will automatically find, and tune to, the largest RF signal. If you know the frequency and level of the signal-under-test, set the HP 892X tuning mode to "manual" and set the input attenuators to "hold". Write your program to directly control the analyzer frequency and input attenuation. To find the correct attenuation, you may want to manually allow the HP 892X to auto-range on a particular signal, then in your automated application you can set the gain correctly under program control.

It is obvious that making unnecessary measurements will take additional time, however many programmers don't realize that their application software is often causing the test set to make unnecessary measurements. There are two situations that cause unnecessary measurements. The first situation is when the displayed screen has more measurements activated than necessary. For instance, the RF Analyzer screen may have Frequency Error, TX Power, FM Deviation, and AF Frequency all active. Whenever a trigger command is issued, the test set will automatically make all the active measurements. You can save a significant amount of time by disabling the unused measurements. A sample of how to program the HP 892X to disable the FM Deviation measurement is:

```
OUTPUT 714;"MEAS:AFR:FM:STATE OFF"
```

A second situation that often generates unnecessary measurements is during repetitive triggering. When making multiple measurements with a single setting, use single triggering, and read back the results from all the measurements. This technique does not cause the HP 892X to re-trigger measurements each time a new query is sent to the test set.

For example:

```
10 ! re-save "c:\hpbasic\892xprog\speed1.pgm"
20 OUTPUT 714;"TRIG:MODE:RETR SINGLE"
30 OUTPUT 714;"TRIG"
40 OUTPUT 714;"MEAS:RFR:POW?"
50 ENTER 714;Pow_result
60 OUTPUT 714;"MEAS:AFR:FM?"
70 ENTER 714;Fm_result
80 PRINT Pow_result,Fm_result
90 END
```

In this example, line 30 triggers all the active measurements, and lines 40 to 70 reads back results.

Another example of essentially the same program is:

```
10 ! re-save "c:\hpbasic\892xprog\speed2.pgm"
20 OUTPUT 714;"TRIG:MODE:RETR SINGLE"
30 OUTPUT 714;"TRIG"
40 OUTPUT 714;"MEAS:RFR:POW?;MEAS:AFR:FM?"
50 ENTER 714;Pow_result,Fm_result
60 PRINT Pow_result,Fm_result
70 END
```

This program saves some HP-IB traffic and reads back multiple measurements by combining two queries and two Enters on the same line. (This technique may be more difficult to implement in languages other than HP BASIC).



For more information about Hewlett-Packard test and measurement products, applications, services, and for a current sales office listing, visit our web site, <http://www.hp.com/go/tmdir>. You can also contact one of the following centers and ask for a test and measurement sales representative.

United States:

Hewlett-Packard Company
Test and Measurement Call Center
P.O. Box 4026
Englewood, CO 80155-4026
1 800 452 4844

Canada:

Hewlett-Packard Canada Ltd.
5150 Spectrum Way
Mississauga, Ontario
L4W 5G1
(905) 206 4725

Europe:

Hewlett-Packard
European Marketing Centre
P.O. Box 999
1180 AZ Amstelveen
The Netherlands
(31 20) 547 9900

Japan:

Hewlett-Packard Japan Ltd.
Measurement Assistance Center
9-1, Takakura-Cho, Hachioji-Shi,
Tokyo 192, Japan
Tel: (81-426) 56-7832
Fax: (81-426) 56-7840

Latin America:

Hewlett-Packard
Latin American Region Headquarters
5200 Blue Lagoon Drive
9th Floor
Miami, Florida 33126
U.S.A.
(305) 267 4245/4220

Australia/New Zealand:

Hewlett-Packard Australia Ltd.
31-41 Joseph Street
Blackburn, Victoria 3130
Australia
1 800 629 485

Asia Pacific:

Hewlett-Packard Asia Pacific Ltd
17-21/F Shell Tower, Times Square,
1 Matheson Street, Causeway Bay,
Hong Kong
Tel: (852) 2599 7777
Fax: (852) 2506 9285